
Overview
    

Diesel is a toolkit that allows you to design DSLs in Elixir.
DSLs built with Diesel are:
	declarative: they look just like HTML
	structured: the schema is checked at compile time
	reusable: the same dsl can be applied to different sets of code generators




  

    
Installation
    

The package can be installed by adding diesel to your list of dependencies in mix.exs:
def deps do
  [
    {:diesel, "~> 0.7"}
  ]
end
Formatter setup
In order to avoid superfluous parenthesis when defining structured tags in your DSL, also add diesel to your dependencies in your .formatter.exs:
[
  inputs: [ ... ],
  import_deps: [..., :diesel],
  locals_without_parens: ...,
]



  

    
Tutorial
    

In this tutorial, we will go through the process of designing a DSL for a finite state machine. We will
use such state machine to express the lifecycle of a payment.

  
    
    A state machine for payments
  


defmodule MyApp.Payments.Payment do
  use MyApp.Fsm

  fsm do
    state :pending do
      on event: :created do
        action SendToGateway
        next state: :sent
      end
    end

    state :sent, timeout: 60 do
      on event: :success do
        action NotifyParties
        next state: :accepted
      end

      on event: :error do
        action NotifyParties
        next state: :declined
      end

      on event: :timeout do
        action NotifyParties
        next state: :declined
      end
    end

    state :accepted do
    end

    state :declined do
    end
  end
end

  
    
    The Fsm library module
  


First, lets define the MyApp.Fsm library module:
defmodule MyApp.Fsm do
  use Diesel,
    otp_app: :my_app,
    dsl: MyApp.Dsm.Dsl  # optional
end
This module will import the api offered by the actual dsl, to be implemented in this example by module MyApp.Fsm.Dsl.
The :dsl key is optional. If omitted, it will default to the caller module, suffixed by Dsl. The above example is equivalent to:
defmodule MyApp.Fsm do
  use Diesel, otp_app: :my_app
end

  
    
    Defining the DSL
  


We will need the following elements of the language:
	fsm: the root tag of the dsl
	state: a definition of a state
	action: an action to be triggered as soon as we enter a state
	on: the definition of an event, in a given state
	next: the next state to transition into

defmodule MyApp.Fsm.Dsl do
  use Diesel.Dsl,
    otp_app: :my_app,
    root: MyApp.Fsm.Dsl.Fsm, # optional
    tags: [
      MyApp.Fsm.Dsl.Action,
      MyApp.Fsm.Dsl.Next,
      MyApp.Fsm.Dsl.On,
      MyApp.Fsm.Dsl.State
    ]
end
The :root key is optional. If omitted, a naming convention will be applied, so that the
above example is equivalent to:
defmodule MyApp.Fsm.Dsl do
  use Diesel.Dsl,
    otp_app: :my_app,
    tags: [
      ...
    ]
end
In the next sections, we will define these as structured tags by relying on the Diesel.Tag built-in dsl.
The fsm root tag
The fsm tag is the root of our DSL. It supports one or many state children tags:
defmodule MyApp.Fsm.Dsl.Fsm do
  use Diesel.Tag

  tag do
    child :state, min: 1
  end
end
The state tag
The state tag requires:
	a name attribute (this is the default attribute name in Diesel). The accepted values are:
pending, sent, accepted, declined.
	an optional timeout attribute
	zero, one or multiple on children

defmodule MyApp.Fsm.Dsl.State do
  use Diesel.Tag

  tag do
    attribute :name, kind: :atom, one_of: [:pending, :sent, :accepted, :declined]
    attribute :timeout, kind: :number, required: false
    child :on, min: 0
  end
end
The :name attribute of any given tag is implicit. For example, the following notation:
state :pending do
  ...
end
is equivalent to:
state name: :pending do
  ...
end
The action tag
The action supports the name of an Elixir module as its only child:
defmodule MyApp.Fsm.Dsl.Action do
  use Diesel.Tag

  tag do
    child kind: :module, min: 1, max: 1
  end
end
When there is a single child involved, the notation
action SomeModule
is equivalent to:
action do
  SomeModule
end
The on tag
The on tag supports:
	the name of the event, as an attribute
	exactly one next child, as the next state
	zero, one or multiple action modules to execute as part of the state transition

defmodule MyApp.Fsm.Dsl.On do
  use Diesel.Tag

  tag do
    attribute :event, kind: :atom
    child :next, min: 0, max: 1
    child :action, min: 0
  end
end
The next tag
The next tag only supports the next state to transition to, as the name attribute:
defmodule MyApp.Fsm.Dsl.Next do
  use Diesel.Tag

  tag do
    attribute :state, kind: :atom, one_of: [:pending, :sent, :accepted, :declined]
  end
end

  
    
    Parsing the DSL
  


We can convert the resulting dsl into an alternative representation, for easier consumption during code generation.
For example, if we define a custom Transition struct to model state transitions:
defmodule MyApp.Fsm.Transition do
  @moduledoc """
  A custom representation of a state transition

  * `from` is the source state
  * `to` is the target state
  * `event` is the triggering event
  * `actions` is a list of Elixir modules to execute as side effects
  """
  defstruct [:from, :to, :event, :actions]
then we can define a parser module that transforms the original raw definition (a tree of tuples) into a plain list of Transition structrs:
defmodule MyApp.Fsm do
  use Diesel,
    otp_app: ...,
    dsl: ...,
    parsers: [
      MyApp.Fsm.Parser
    ]
end
Please check the Fsm.Parser module included in test/support/fsm.ex.
The :parsers key is optional. If omitted, a default parser will be used, by appending
the Parser suffix to the caller module. The above example is equivalent to:
defmodule MyApp.Fsm do
  use Diesel,
    otp_app: ...,
    dsl: ...,
end
In reality, parsers are optional. If you wish to skip them entirely, you can set an empty list:
defmodule MyApp.Fsm do
  use Diesel,
    otp_app: ...,
    dsl: ...,
    parsers: []
end

  
    
    Generating code
  


Once our state machine is parsed into a list of transitions, we can then generate any custom code of our choice and inject it into our MyApp.Fsm module.
Generated code is provided by implementations of the Diesel.Generator behaviour. A generator returns one or more Elixir quoted expressions from its generate/2 callback.
For example, in order to generate a diagram/0 function that returns a Graphviz diagram for our state machine, we could make use of module Fsm.Diagram, also included in test/support/fsm.ex:
defmodule MyApp.Fsm do
  use Diesel,
    otp_app: ...,
    dsl: ...,
    parsers: ...,
    generators: [
      Fsm.Diagram
    ]
end



  

    
Parsers
    


  
    
    Raw definitions
  


By default, DSLs generated with Diesel are compiled into a tree-like structure of nodes as tuples.
This is similar to the structure you'd expect from a HTML document parsed with Loki. For example,
the following code:
defmodule MyApp.Fsm.Paymnent do
  use MyApp.Fsm

  fsm do
    state :pending do
    end
  end
end
gives the following internal raw definition:
{:fsm, [], [
  {:state, [name: :pending], []}
]}
By default, this is the datastructure that will be then consumed by code generators.

  
    
    The Parser behaviour
  


It is possible however to add parsing steps and convert the raw definition into a more suitable data
structure, before it is consumed by generators.
All you need to do is implement the Diesel.Parser behaviour in a new elixir module:
defmodule MyApp.Fsm.MyParser do
  @moduledoc "Converts a fsm definition into structs"
  @behaviour Diesel.Parser

  @impl true
  def parse({:fsm, [], states}, _opts) do
    %Fsm{states: [...]}
  end
end
and add it to the list of parsers:
defmodule MyApp.Fsm do
  use Diesel,
    otp_app: ...,
    dsl: ...,
    parsers: [
      MyApp.Fsm.MyParser
    ]

  defstruct states: []
end



  

    
Generators
    

By implementing the Diesel.Generator behaviour, developers can produce elixir code, based on a DSL
definition. For example, we could imagine a GenServer based implementation of a state machine:
defmodule MyApp.Fsm do
  use Diesel,
    otp_app: ...,
    dsl: ...,
    parsers: [...],
    generators: [
      MyApp.Fsm.Generator.GenServer
    ]
where:
defmodule MyApp.Fsm.Generator.GenServer do
  @moduledoc "Generates a GenServer for a state machine DSL"
  @behaviour Diesel.Generator

  @impl true
  def generate(definition, _opts) do
    quote do
      use GenServer
      ...
    end
  end
end



  

    
Unstructured tags
    

In general, it is highly recommended to express DSLs in terms of structured tags, by using the
Diesel.Tag macro. This will enforce a well defined schema and the compiler will be able to pick up
misconfigurations and/or omissions and raise errors early.
This said, it is also possible to express a DSL in terms of unstructured, or plain atom tags, when
more informal definitions might be acceptable:
defmodule MyApp.Html.Dsl do
  use Diesel.Dsl,
    otp_app: :my_app,
    root: :html,
    tags: [:head, :body, :meta, :script, ...]
end



  

    
Kernel conflicts
    

Depending on how you define your DSL, you might get compiler errors in the form function /Y imported from both (YOUR DSL) and Kernel, call is ambiguous.
You can stop importing the one from Kernel via the :overrides key, eg:
defmodule MyApp.Html do
  use Diesel,
    otp_app: :my_app,
    dsl: MyApp.Html.Dsl,
    overrides: [div: 1]



  

    
Debug
    

You can turn debug on, in order to see the code being generated by Diesel at compile time.
defmodule MyApp.MyFsm do
  use Diesel, debug: true
  ...
end



  

    
Diesel behaviour
    



      
Declarative programming in Elixir
Diesel is a toolkit that helps you build your own DSLs.
Usage:
defmodule MyApp.Fsm do
  use Diesel,
    otp_app: :my_app,
    dsl: MyApp.Fsm.Dsl,
    parsers: [
      ...
    ],
    generators: [
      ...
    ]
end
For more information on how to use this library, please check:
	the Diesel.Dsl and Diesel.Tag modules,
	the guides and tutorials provided in the documentation
	the examples used in tests

          @callback definition() :: element()


      


Returns the raw definition for the dsl, before compilation

  


        

      



  

    
Diesel.Dsl 
    



      
Defines the syntax provided by a DSL.
Simple usage:
defmodule MyApp.Fsm.Dsl do
  use Diesel.Dsl,
    root: :fsm,
    tags: [
      :state,
      :on,
      :action,
      :next
    ]
end
Please check the documentation for more info on how to extend DSLs via packages

      





  

    
Diesel.Generator behaviour
    



      
Generators produce actual Elixir code from DSLs.
Given a definition, a generator returns one or multiple quoted expressions.
The following info are passed in the keyword list of options:
	:otp_app
	:caller_module

          @callback generate(definition :: term(), opts :: Keyword.t()) :: Macro.t() | [Macro.t()]

Naming utilities

Diesel.Parser behaviour
    



      
A parser is a DSL tranformation step before code generation
Example:
def parse({:music, attrs, children}, _opts) do
  # eg return some struct here
end
Parsing a definition is an optional step.

Returns a built-in parser, given its name
Supported names:
	strip_root


  


        

      



  

    
Diesel.Tag 
    



      
Tags are the building blocks for the syntax offered by a Diesel DSL.
Usage:
defmodule MyApp.Fsm.Dsl.State do
  use Diesel.Tag

  attribute :name, kind: :atom
  child :on, :min: 0
end
For more information on how to define structured tags, please check the examples provided in the documentation and in tests

Returns the name of the given tag
For unstructured, plain atom tags, the name is simply the atom itself

Validates the given definition node against the given tag
This function supports both structured and plain atom tags.
For plain atom tags, this function will always return success.

  


        

      



  

    
Diesel.Tag.Dsl 
    



      
The meta-dsl that allow developers to design structured tags

Returns the list of locals without parens function signatures, so that they can be easily
included in .formatter.exs files

Helper module that validates a tag against a schema.
Validation of tags is only available for structured tags, and is applied right after tag parsing
and before code generation.

          @spec validate(node :: tuple(), schema :: tuple()) :: :ok | {:error, any()}


      


Validates the given node, against the given schema

